NAG Library Routine Document

D01GZF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D01GZF calculates the optimal coefficients, for use by D01GCF and D01GDF, when the number of points is the product of two primes.

2 Specification

```
SUBROUTINE D01GZF(NDIM, NP1, NP2, VK, IFAIL)
INTEGER           NDIM, NP1, NP2, IFAIL
double precision  VK(NDIM)
```

3 Description

Korobov (1963) gives a procedure for calculating optimal coefficients for p-point integration over the n-cube $[0,1]^n$, when the number of points is

$$p = p_1 p_2 \tag{1}$$

where p_1 and p_2 are distinct prime numbers.

The advantage of this procedure is that if p_1 is chosen to be the nearest prime integer to p_2^2, then the number of elementary operations required to compute the rule is of the order of $p^{4/3}$ which grows less rapidly than the number of operations required by D01GYF. The associated error is likely to be larger although it may be the only practical alternative for high values of p.

4 References

Korobov N M (1963) *Number Theoretic Methods in Approximate Analysis* Fizmatgiz, Moscow

5 Parameters

1: NDIM – INTEGER *Input*

On entry: n, the number of dimensions of the integral.

Constraint: NDIM ≥ 1.

2: NP1 – INTEGER *Input*

On entry: the larger prime factor p_1 of the number of points in the integration rule.

Constraint: NP1 must be a prime number ≥ 5.

3: NP2 – INTEGER *Input*

On entry: the smaller prime factor p_2 of the number of points in the integration rule. For maximum efficiency, p_2^2 should be close to p_1.

Constraint: NP2 must be a prime number such that NP1 $>$ NP2 ≥ 2.

4: VK(NDIM) – ***double precision*** array *Output*

On exit: the n optimal coefficients.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NDIM < 1.

IFAIL = 2

On entry, NP1 < 5,
or NP2 < 2,
or NP1 ≤ NP2.

IFAIL = 3

The value NP1 × NP2 exceeds the largest integer representable on the machine, and hence the optimal coefficients could not be used in a valid call of D01GCF.

IFAIL = 4

On entry, NP1 is not a prime number.

IFAIL = 5

On entry, NP2 is not a prime number.

IFAIL = 6

The precision of the machine is insufficient to perform the computation exactly. Try smaller values of NP1 or NP2, or use an implementation with higher precision.

7 Accuracy

The optimal coefficients are returned as exact integers (though stored in a ***double precision*** array).

8 Further Comments

The time taken by D01GZF grows at least as fast as $(p_1 p_2)^{4/3}$. (See Section 3.)

9 Example

This example calculates the Korobov optimal coefficients where the number of dimensons is 4 and the number of points is the product of the two prime numbers, 89 and 11.

9.1 Program Text

```
*     D01GZF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NDIM
      PARAMETER        (NDIM=4)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, NP1, NP2
*     .. Local Arrays ..
      DOUBLE PRECISION VK(NDIM)
*     .. External Subroutines ..
      EXTERNAL         D01GZF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D01GZF Example Program Results'
      NP1 = 89
      NP2 = 11
      WRITE (NOUT,*)
      IFAIL = 1
*
      CALL D01GZF(NDIM,NP1,NP2,VK,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,99999) 'NDIM =', NDIM, ' NP1 =', NP1, ' NP2 =', NP2
         WRITE (NOUT,*)
         WRITE (NOUT,99998) 'Coefficients =', (VK(I),I=1,NDIM)
      ELSE
         WRITE (NOUT,99997) ' ** D01GZF returned with IFAIL = ', IFAIL
      END IF
*
99999 FORMAT (1X,A,I3,A,I6,A,I6)
99998 FORMAT (1X,A,4F6.0)
99997 FORMAT (1X,A,I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D01GZF Example Program Results

 NDIM =  4 NP1 =    89 NP2 =    11

 Coefficients =    1.  102.  614.  951.
```